
Overview
This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 8.1.4.1.5 which is part of the NR5GC_IWD_20wk25 test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 3). Execution log files are provided as evidence. 

Contact:
Ting Gao

gaoting@starpointcomm.com
Verification Test Summary

Test Case: 
8.1.4.1.5
ATS Version:
iwd-TTCN3-B2019-06_D20wk25
System Simulator used:
Starpoint SP9500 Test system
UE used:
HiSilicon Balong 5000 
Verification Status:
PASS
Corrections required

Note: Change2 in R5s201240 is also applicable for this test case.
Change 1
	Function name
	fl_TC_8_1_4_1_5_TestBody

	Reason for change
	At Step 2, f_NR5GC_RRCReconfiguration_IntraNR_HO will release nr_Cell1 SRBs and DRBs. 

According to 5.3.7.2 in 38.331, MAC reset, spcell config released, so RRCReestablishmentComplete must be sent through the rach process, and SearchSpaceType in DL and UL SearchSpaceAndDci is set to cssType1.
At step 10,the value of v PhysCellId is incorrect.The Physical Cell Identity of the PCell the UE was connected to prior to the failure.


	Summary of change
	Reestablish SS SRBs and DRBs for nr_Cell1.
Add CBRA_crnti based rach configuration before sending RRC Reestablishment.
Set the value of v PhysCellId to PhysCellId of cell 1.

	TTCN module
	RRC_IntraNR_Handover_NR5GC.ttcn

	MCC160 Comment
	


Before change

   function fl_TC_8_1_4_1_5_TestBody () runs on NR5GC_PTC

  {

    var RNTI_Value v_RNTI_Value := f_NR_CellInfo_GetRNTI(nr_Cell1);

    var PhysCellId v_PhysCellId := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    var ShortMAC_I v_ShortMAC_I;

    var template (value) DL_DCCH_Message v_RRCReconfigurationHO;

    var NextHopChainingCount v_NextHopChainingCount;

    var float v_T304Min := f_NR_SetTimerToleranceMin(nr_Cell1, rrcTimer, 1.0); //T304 is ms1000 according to 38.508-1 Table 4.6.3-19.

    var float v_T304Max := f_NR_SetTimerToleranceMax(nr_Cell1, rrcTimer, 1.0); //T304 is ms1000 according to 38.508-1 Table 4.6.3-19.

    var template(value) TimingInfo_Type v_TimingInfo;

    var SubFrameTiming_Type v_SubframeTiming;

    //Power level as per Table 8.1.4.1.5.3.2-1

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1 := { cs_NR_CellPower(nr_Cell2, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)};

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT2 := { cs_NR_CellPower(nr_Cell2, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)};

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT3 := { cs_NR_CellPower(nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),

                                                                      cs_NR_CellPower(nr_Cell2, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)};

    timer t_T304Min;

    // Reconfigure nr_Cell2 to no respond to any RA Preamble

    // Table 8.1.4.1.5.3.2-4: The UE attempts to perform the handover using MAC Random Access Preamble on NR Cell 2. SS does not respond.

    f_NR_SS_RachProcedureConfig(nr_Cell2, cs_NR_RachProcedureConfig_NoResponse);

    //@siclog "Step 1" siclog@

    // The SS changes the power level setting according to the row "T1".

    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 2" siclog@

    // The SS transmits an RRCReconfiguration message on NR Cell 1 to order the UE to perform handover to NR Cell 2.

    // Note: failure case UE doesn't go to nr_Cell2 => the standard function need not be used

    v_RRCReconfigurationHO := f_NR5GC_RRCReconfiguration_IntraNR_HO(nr_Cell2);

    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, cs_TimingInfo_Now, v_RRCReconfigurationHO));

    t_T304Min.start(v_T304Min);

    //@siclog "Step 3" siclog@

    //The SS changes the power level setting according to the row "T2".

    f_NR_SetCellPowerList(v_CellPowerList_AtT2);

    //@siclog "Step 4" siclog@

    //Check: Does the UE transmit an RRCReestablishmentRequest message on NR Cell 1?

    v_ShortMAC_I := f_NR_Calculate_ShortMAC(nr_Cell1, nr_Cell1);

    t_T304Min.timeout;

    SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCReestablishmentRequest(cr_NR_ReestabUE_Identity (v_RNTI_Value, v_PhysCellId, v_ShortMAC_I), handoverFailure)));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    v_NextHopChainingCount := 2;

    f_NR_SS_AS_ActivateSecurity_Current(nr_Cell1, v_NextHopChainingCount);

    //@siclog "Step 5" siclog@

    //The SS transmits an RRCReestablishment message to resume SRB1 operation and re-activate security on NR Cell 1.

    SRB.send (cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, cs_TimingInfo_Now, cs_38508_RRCReestablishment_Def(tsc_NR_RRC_TI_Def,v_NextHopChainingCount)));

    //@siclog "Step 6" siclog@

    // Does the UE transmit an RRCReestablishmentComplete message using the security key derived from the nextHopChainingCount on NR Cell 1?

    SRB.receive ( car_NR_SRB1_RrcPdu_IND ( nr_Cell1,  cr_38508_RRCReestablishmentComplete) );

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    //@siclog "Steps 6A/6B" siclog@

    //The SS transmits an RRCReconfiguration message to resume existing radio bearers on NR Cell 1.

    //The UE transmits an RRCReconfigurationComplete message on NR Cell 1.

    //@sic R5-198924 sic@

    f_NR_RRCReconfigExistingDRBs(nr_Cell1, omit);

    //@siclog "Step 7" siclog@

    //The SS changes the power level setting according to the row "T1".

    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 8" siclog@

    //The SS transmits an RRCReconfiguration message on NR Cell 1 to order the UE to perform handover to NR Cell 2.

    v_SubframeTiming := f_NR_GetNextSendOccasion_DL(nr_Cell1);

    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1,  cs_TimingInfo_NR(v_SubframeTiming), v_RRCReconfigurationHO));

   //Re-configure RACH procedure as for initial access in nr_Cell2

    v_TimingInfo := f_NR_TimingOtherCell(nr_Cell1, cs_TimingInfo_NR(v_SubframeTiming), nr_Cell2);

    v_SubframeTiming := f_SubFrameTiming_AddMilliSeconds(valueof(v_TimingInfo.SubFrame), float2int(v_T304Max * 1000.0)); //Add T304max to make sure T304 timeout

    f_NR_SS_RachProcedureConfig_Def(nr_Cell2,  cs_TimingInfo_NR(v_SubframeTiming));

    //EXCEPTION: In parallel to the events described in step 9 the steps specified in Table 8.1.4.1.5.3.2-4 should take place.

    //Nothing to do nr_Cell2 already configured to not respond to any RA Preamble

    //@siclog "Step 9" siclog@

    //The SS changes the power level setting according to the row "T3".

    f_NR_SetCellPowerList(v_CellPowerList_AtT3);

    //@siclog "Step 10" siclog@

    // Check: Does the UE transmit an RRCReestablishmentRequest message on NR Cell 2?

    v_ShortMAC_I := f_NR_Calculate_ShortMAC(nr_Cell1, nr_Cell2);

    v_PhysCellId := f_NR_CellInfo_GetPhysicalCellId(nr_Cell2);
    SRB.receive(car_NR_SRB0_RrcPdu_IND ( nr_Cell2, cr_38508_RRCReestablishmentRequest(cr_NR_ReestabUE_Identity (v_RNTI_Value, v_PhysCellId, v_ShortMAC_I), handoverFailure)));

    v_NextHopChainingCount := 0;

    f_NR_SS_AS_ActivateSecurity_Current ( nr_Cell2, v_NextHopChainingCount);

    //@siclog "Step 11" siclog@

    //The SS transmits an RRCReestablishment message to resume SRB1 operation and re-activate security on NR Cell 2.

    SRB.send (cas_NR_SRB1_RrcPdu_REQ ( nr_Cell2, cs_TimingInfo_Now, cs_38508_RRCReestablishment_Def(tsc_NR_RRC_TI_Def,v_NextHopChainingCount)));

    //@siclog "Step 12" siclog@

    // Does the UE transmit an RRCReestablishmentComplete message using the security key derived from the nextHopChainingCount on NR Cell 2?

    SRB.receive ( car_NR_SRB1_RrcPdu_IND ( nr_Cell2,  cr_38508_RRCReestablishmentComplete) );

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12");

    //@siclog "Steps 12A/12B" siclog@

    //The SS transmits an RRCReconfiguration message to resume existing radio bearers on NR Cell 2.

    //The UE transmits an RRCReconfigurationComplete message on NR Cell 2.

    //@sic R5-198924 sic@

    f_NR_RRCReconfigExistingDRBs(nr_Cell2, omit, -);

    //Release cell configuration in nr_Cell1 is not needed as it is the end of the test

    //@siclog "Step 13" siclog@

    //The SS transmits an RRCRelease message on NR Cell 2.

    f_NR_RRCRelease(nr_Cell2);

  }

After change

  function fl_TC_8_1_4_1_5_TestBody () runs on NR5GC_PTC

  {

    var RNTI_Value v_RNTI_Value := f_NR_CellInfo_GetRNTI(nr_Cell1);

    var PhysCellId v_PhysCellId := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    var ShortMAC_I v_ShortMAC_I;

    var template (value) DL_DCCH_Message v_RRCReconfigurationHO;

    var NextHopChainingCount v_NextHopChainingCount;

    var float v_T304Min := f_NR_SetTimerToleranceMin(nr_Cell1, rrcTimer, 1.0); //T304 is ms1000 according to 38.508-1 Table 4.6.3-19.

    var float v_T304Max := f_NR_SetTimerToleranceMax(nr_Cell1, rrcTimer, 1.0); //T304 is ms1000 according to 38.508-1 Table 4.6.3-19.

    var template(value) TimingInfo_Type v_TimingInfo;

    var SubFrameTiming_Type v_SubframeTiming;

    //Power level as per Table 8.1.4.1.5.3.2-1

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1 := { cs_NR_CellPower(nr_Cell2, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)};

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT2 := { cs_NR_CellPower(nr_Cell2, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)};

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT3 := { cs_NR_CellPower(nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),

                                                                      cs_NR_CellPower(nr_Cell2, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)};

    timer t_T304Min;

    var NR_PhysicalParameters_Type v_NR_PhysicalParameters;

    var template(value) NR_RachProcedureConfig_Type v_NR_RachProcedure; 

    var IntegerList_Type v_DrbIdList := f_NR_GetActiveDRBs(nr_Cell1); 

    // Reconfigure nr_Cell2 to no respond to any RA Preamble

    // Table 8.1.4.1.5.3.2-4: The UE attempts to perform the handover using MAC Random Access Preamble on NR Cell 2. SS does not respond.

    f_NR_SS_RachProcedureConfig(nr_Cell2, cs_NR_RachProcedureConfig_NoResponse);

    //@siclog "Step 1" siclog@

    // The SS changes the power level setting according to the row "T1".

    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 2" siclog@

    // The SS transmits an RRCReconfiguration message on NR Cell 1 to order the UE to perform handover to NR Cell 2.

    // Note: failure case UE doesn't go to nr_Cell2 => the standard function need not be used

    v_RRCReconfigurationHO := f_NR5GC_RRCReconfiguration_IntraNR_HO(nr_Cell2);

    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, cs_TimingInfo_Now, v_RRCReconfigurationHO));

    t_T304Min.start(v_T304Min);

    //@siclog "Step 3" siclog@

    //The SS changes the power level setting according to the row "T2".

    f_NR_SetCellPowerList(v_CellPowerList_AtT2);

    //@siclog "Step 4" siclog@

    //Check: Does the UE transmit an RRCReestablishmentRequest message on NR Cell 1?

    v_ShortMAC_I := f_NR_Calculate_ShortMAC(nr_Cell1, nr_Cell1);

    t_T304Min.timeout;

    SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCReestablishmentRequest(cr_NR_ReestabUE_Identity (v_RNTI_Value, v_PhysCellId, v_ShortMAC_I), handoverFailure)));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    f_NR_SS_SRBs_DRBs_Release(nr_Cell1, cs_TimingInfo_Now, v_DrbIdList); 

    f_NR_SS_SRBs_DRBs_Config(nr_Cell1, cs_TimingInfo_Now, omit, true); 

    v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(nr_Cell1);

    v_NR_RachProcedure := cs_NR_RachProcedureConfig_CBRA_CrntiBased(v_NR_PhysicalParameters, -, tsc_C_RNTI_Value1);


  // configure CBRA CRNTI Based RACH procedure

      SYS.send(cads_NR_RachProcedureConfig_DcchDtchConfig_REQ(nr_Cell1, cs_TimingInfo_Now, v_NR_RachProcedure,v_NR_PhysicalParameters));



      SYS.receive(car_NR_CellConfig_CNF(nr_Cell1)); 

    v_NextHopChainingCount := 2;

    f_NR_SS_AS_ActivateSecurity_Current(nr_Cell1, v_NextHopChainingCount);

    //@siclog "Step 5" siclog@

    //The SS transmits an RRCReestablishment message to resume SRB1 operation and re-activate security on NR Cell 1.

    SRB.send (cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, cs_TimingInfo_Now, cs_38508_RRCReestablishment_Def(tsc_NR_RRC_TI_Def,v_NextHopChainingCount)));

    //@siclog "Step 6" siclog@

    // Does the UE transmit an RRCReestablishmentComplete message using the security key derived from the nextHopChainingCount on NR Cell 1?

    SRB.receive ( car_NR_SRB1_RrcPdu_IND ( nr_Cell1,  cr_38508_RRCReestablishmentComplete) );

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    //@siclog "Steps 6A/6B" siclog@

    //The SS transmits an RRCReconfiguration message to resume existing radio bearers on NR Cell 1.

    //The UE transmits an RRCReconfigurationComplete message on NR Cell 1.

    //@sic R5-198924 sic@

    f_NR_RRCReconfigExistingDRBs(nr_Cell1, omit,-,-,-,-,-,-,-,-,true);

    //@siclog "Step 7" siclog@

    //The SS changes the power level setting according to the row "T1".

    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 8" siclog@

    //The SS transmits an RRCReconfiguration message on NR Cell 1 to order the UE to perform handover to NR Cell 2.

    v_SubframeTiming := f_NR_GetNextSendOccasion_DL(nr_Cell1);

    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1,  cs_TimingInfo_NR(v_SubframeTiming), v_RRCReconfigurationHO));

   //Re-configure RACH procedure as for initial access in nr_Cell2

    v_TimingInfo := f_NR_TimingOtherCell(nr_Cell1, cs_TimingInfo_NR(v_SubframeTiming), nr_Cell2);

    v_SubframeTiming := f_SubFrameTiming_AddMilliSeconds(valueof(v_TimingInfo.SubFrame), float2int(v_T304Max * 1000.0)); //Add T304max to make sure T304 timeout

    f_NR_SS_RachProcedureConfig_Def(nr_Cell2,  cs_TimingInfo_NR(v_SubframeTiming));

    //EXCEPTION: In parallel to the events described in step 9 the steps specified in Table 8.1.4.1.5.3.2-4 should take place.

    //Nothing to do nr_Cell2 already configured to not respond to any RA Preamble

    //@siclog "Step 9" siclog@

    //The SS changes the power level setting according to the row "T3".

    f_NR_SetCellPowerList(v_CellPowerList_AtT3);

    //@siclog "Step 10" siclog@

    // Check: Does the UE transmit an RRCReestablishmentRequest message on NR Cell 2?

    v_ShortMAC_I := f_NR_Calculate_ShortMAC(nr_Cell1, nr_Cell2);

    v_PhysCellId := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

cr_38508_RRCReestablishmentRequest(cr_NR_ReestabUE_Identity (v_RNTI_Value, v_PhysCellId, v_ShortMAC_I), handoverFailure)));

    v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(nr_Cell2);

    v_NR_RachProcedure := cs_NR_RachProcedureConfig_CBRA_CrntiBased(v_NR_PhysicalParameters, -, tsc_C_RNTI_Value1);


  // configure CBRA CRNTI Based RACH procedure

    SYS.send(cads_NR_RachProcedureConfig_DcchDtchConfig_REQ(nr_Cell2, cs_TimingInfo_Now, v_NR_RachProcedure,v_NR_PhysicalParameters));



    SYS.receive(car_NR_CellConfig_CNF(nr_Cell2)); 

    v_NextHopChainingCount := 0;

    f_NR_SS_AS_ActivateSecurity_Current ( nr_Cell2, v_NextHopChainingCount);

    //@siclog "Step 11" siclog@

    //The SS transmits an RRCReestablishment message to resume SRB1 operation and re-activate security on NR Cell 2.

    SRB.send (cas_NR_SRB1_RrcPdu_REQ ( nr_Cell2, cs_TimingInfo_Now, cs_38508_RRCReestablishment_Def(tsc_NR_RRC_TI_Def,v_NextHopChainingCount)));

    //@siclog "Step 12" siclog@

    // Does the UE transmit an RRCReestablishmentComplete message using the security key derived from the nextHopChainingCount on NR Cell 2?

    SRB.receive ( car_NR_SRB1_RrcPdu_IND ( nr_Cell2,  cr_38508_RRCReestablishmentComplete) );

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12");

    //@siclog "Steps 12A/12B" siclog@

    //The SS transmits an RRCReconfiguration message to resume existing radio bearers on NR Cell 2.

    //The UE transmits an RRCReconfigurationComplete message on NR Cell 2.

    //@sic R5-198924 sic@

    f_NR_RRCReconfigExistingDRBs(nr_Cell2, omit);

    //Release cell configuration in nr_Cell1 is not needed as it is the end of the test

    //@siclog "Step 13" siclog@

    //The SS transmits an RRCRelease message on NR Cell 2.

    f_NR_RRCRelease(nr_Cell2);

  } 
Execution Log Files 

HiSilicon Balong 5000
The HiSilicon Balong 5000 passed this test case on Starpoint SP9500 Test System. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
\ TC_8_1_4_1_5_NR5GC_PASS.spm
  PIXCIT settings used:

 \ NR5GC_Testsuite.sppar

	R5s201243:  Supporting information for agreement of NR5GC test case 8.1.4.1.5 in FR1


